Mechanising Hall's Theorem for Countable Graphs

Fabián Fernando Serrano Suárez[1], Mauricio Ayala-Rincón[2], and Thaynara Arielly de Lima[3]

[1] Universidad Nacional de Colombia - Sede Manizales
ffserranos@unal.edu.co
[2] Universidade de Brasília
ayala@unb.br
[3] Universidade Federal de Goiás
thaynaradelima@ufg.br

Abstract

This work presents a formalisation in Isabelle/HOL of the extension of Hall's theorem for finite graphs to countable graphs. The proof uses a formalisation of the author's countable set-theoretical version of Hall's theorem that was proved as a consequence of the marriage-condition characterisation for finite families of sets and a formalisation in Isabelle/HOL of the compactness theorem for propositional logic. The development is a first step toward mechanising infinite versions of results equivalent to Hall's marriage theorem in contexts other than set theory.

1 Introduction

Hall's marriage theorem is a landmark result established primarily by Richard Hall [12], and it is equivalent to several other significant theorems in combinatorics and graph theory (cf. [3], [4], [21]), namely: Menger's theorem (1929), König's minimax theorem (1931), König–Egerváry theorem (1931), Dilworth's theorem (1950), Max Flow-Min Cut theorem (related to Ford-Fulkerson algorithm), among others. Consequently, any mechanisation of Hall's theorem allows one to formally prove any of those equivalent results.

Two well-known versions of Hall's theorem exist, one for finite families of finite sets and another for finite graphs. The proofs of any previously cited equivalences can be more adapted to a specific version of Hall's Theorem, either the set-theoretical or the graph-theoretical version. For example, König–Egerváry theorem states that the minimum cover in a finite bipartite graph has the same cardinality as a maximum matching. If we assume Hall's theorem for finite graphs, one way to infer König–Egerváry theorem is by building a reduction from the latter to the former. Considering the nature of König–Egerváry theorem, it is clear that the graph-theoretical version of Hall's theorem is more appropriate than the set version to establish the equivalence between these theorems.

Although we referred to the finite versions of the mentioned results in the previous paragraphs, we point out that extensions to infinite sets and graphs are of primary interest [2].

Mechanisations such as those presented in this work aim to pave the way to develop formalisations of infinite versions of some theorems in combinatorics related to Hall's Theorem. For example, the authors formalised the set-theoretical version of Hall's Theorem for a countable collection of finite subsets $\{S_i\}_{i \in I}$ of a set S [25]. Such a development applied a formalisation of the compactness theorem for propositional logic, developed by Serrano in [24], and the formalisation of Jiang and Nipkow's for the finite case of the set-theoretical version of Hall's theorem [16].

This work discusses how, applying authors development in [25], the infinite graph-theoretical version of Hall's theorem is mechanised in Isabelle/HOL. The result applies to a general class of infinite bipartite graphs with finite neighbourhoods regarding one of the sets of vertices of the vertex bipartition. The utility of this formalisation is crucial since it can be applied to establish mechanisations of other combinatorial results on infinite sets and graphs, such as variants of Dilworth's theorem, Max Flow-Min Cut theorem, and particular versions of König–Egerváry theorem.

Interestingly other combinatorial well-known results equivalent to Hall's theorem in the finite case are not straightforwardly equivalent in the infinite case; for instance, the infinite version of König-Egerváry theorem that as reported in [2] cannot be inferred from the compactness theorem.

The paper is organised as follows. Section 2 discusses Hall's marriage theorem for finite and infinite countable sets and graphs and explains the equivalence between the version for graphs and sets. Then, Section 3 presents the formalisation in Isabelle/HOL of the graph-theoretical version of Hall's theorem for countable graphs. Section 4 discusses related work before concluding in Section 5.

For the benefit of the reviewing process, the paper includes links to the formalisation highlighted by the symbol ↗.

2 Hall's Theorem for sets and graphs

2.1 Finite and infinite versions of Hall's theorem

Hall's theorem for sets establishes that a finite family $\{S_i\}_{i\in I}$ of finite sets not necessarily disjoint, of elements in a set S, has a system of distinct representatives (SDR) if and only if the so called *marriage condition* holds. The marriage condition states that:

$$\text{For any } J \subseteq I, |J| \leq |\bigcup_{j\in J} S_j|$$

Above, an SDR for the family $\{S_i\}_{i\in I}$ is understood as a subset of elements of S that contains exactly an element for each set in the family. This can be formalised as an injective function $f : I \to S$, such that $f(i) \in S_i$, for $i \in I$.

Definition 1 (SDR)**.** *Let S be an arbitrary set and $\{S_i\}_{i\in I}$ a collection of not necessarily distinct subsets of S with indices in the set I. An injective function $f : I \to \bigcup_{i\in I} S_i$ is an SDR for $\{S_i\}_{i\in I}$ if for all $i \in I$, $f(i) \in S_i$.*

Using the compactness theorem, a proof of a countable infinite version of this theorem was formalised in Isabelle/HOL [25]. The infinite version states that a countable family of finite sets has a set of distinct representatives if and only if the marriage condition below holds:

$$\text{For any } J \subseteq I, J \text{ finite}, |J| \leq |\bigcup_{j\in J} S_j|$$

Above, I is any countable set.

The Hall's theorem for finite graphs states that in a bipartite graph $G = \langle X, Y, E\rangle$ there is a perfect matching covering X if and only if $|J| \leq |N(J)|$ for all $J \subseteq X$. In a bipartite graph with $E \subseteq X \times Y$ and $x \in X \cup Y$, the neighbourhood of x is the set of vertices $N(x) = \{y \mid (x, y) \in E, \text{ or } (y, x) \in E\}$. N is extended in a straightforward manner to sets.

Definition 2 (Directed bipartite digraph and perfect matching)**.** *Let X and Y be non-empty sets. The triple $G = \langle X, Y, E\rangle$ is a directed bipartite digraph if and only if the following conditions hold.*

1. $X \cap Y = \emptyset$, *and*

2. $E \subseteq (X \times Y)$.

A subset of arcs $E' \subseteq E$ *is a perfect matching of* $G = \langle X, Y, E \rangle$ *if and only if*

1. $X = \{x \mid (x, y) \in E'\}$, *and*

2. *if* $(x, y_1), (x, y_2) \in E'$ *then* $y_1 = y_2$*, and if* $(x_1, y), (x_2, y) \in E'$ *then* $x_1 = x_2$.

The infinite version of Hall's theorem for graphs states that in a countable bipartite graph $G = \langle X, Y, E \rangle$, where for all $x \in X$, $N(x)$ is finite, there is a perfect matching covering X if and only if $|J| \leq |N(J)|$ for all J finite, $J \subseteq X$.

Notice that for the infinite version of this theorem the finiteness of $N(x)$ cannot be relaxed; in fact, the graph $G = \langle \mathbb{N}, \mathbb{N}^+, \{(0, i) \mid i \in \mathbb{N}^+\} \bigcup \{(i, i) | i \in \mathbb{N}^+\} \rangle$ is an easy counterexample. In G, the sets of vertices $\mathbb{N}$ and $\mathbb{N}^+$ are seen as different copies of natural numbers.

The formalisation in Isabelle/HOL of the countable version of of Hall's Theorem for sets in [25] uses Nipkow's formalisation of Hall's theorem for finite sets [15] and Serrano's formalisation of the compactness theorem for propositional logic in [24].

2.2 Countable versions of Hall's theorem for sets and graphs

The relation between both countable versions of this theorem for sets and graphs is clear intuitively.

On the one side, a countable bipartite graph $G = \langle X, Y, E \rangle$ gives a countable family of neighbourhoods $\{N(x)\}_{x \in X}$, which are finite sets under the constraint that neighbourhoods of vertices in X are finite. If M is a perfect matching of G, thus one builds an SDR by considering the injective function $f : X \to Y$ such that, for each $x \in X$, $f(x) = y$, where $(x, y) \in M$.

On the other side, if one has a countable family of finite sets $\{S_i\}_{i \in I}$ satisfying the marriage condition, then there exists a distinct set of representatives for $\{S_i\}_{i \in I}$, given by f. We consider the countable bipartite graph built as $G = \langle I, \bigcup_{i \in I} S_i, E \rangle$, where $E = \{(i, y) \mid i \in I, y \in S_i\}$. Since the sets in the countable family of sets $\{S_i\}_{i \in I}$ are finite the set of neighbourhoods in G, for each $i \in I$, $N(i)$, is finite; indeed, $|S_i| = |N(i)|$. The perfect matching covering I is given by the set of arcs $M = \{(i, f(i)) \mid i \in I\}$. By the injectivity of f, pairs of arcs in M have no vertices in common.

The reported formalisation refers to the mechanisation of Hall's theorem for graphs, as a consequence of the set version.

3 Formalisation of Hall's Theorem for Graphs

The formalisation depends on reductions from infinite families of sets to infinite bipartite graphs. Initially, we discuss reductions from families of sets to graphs and vice versa and how the proposed development guarantees correct constructions of matchings from SDRs and vice versa. Then, we explain how the proof of correction of the specialised construction of an SDR from a perfect matching over an infinite directed bipartite graph is used to conclude the infinite graph-theoretical version of Hall's theorem.

3.1 From sets to graphs and vice versa - formalisation of reductions

The formalisation is constructive, and the kernel of it is the transformations of indexed infinite families of sets to and from directed bipartite digraphs. One of the vital features of our formalisation is how we build a *system of distinct representatives* (SDR) for a family of sets from a perfect matching over arbitrary directed bipartite graphs. Such transformations are more general than those discussed in the previous section since neither the family of sets need to be countable nor the sets in the family must be restricted to finite sets. Thus, the bipartite graph may also be non-countable, and the neighbourhoods of the vertices do not need to be finite. Theorems 1 and 2 present the reductions from a problem to another one.

Theorem 1 (SDR associated to a directed bipartite digraph)**.** *Let $G = \langle X, Y, E\rangle$ be a directed bipartite digraph.*

The collection of sets associated to G is built as $\{V_i\}_{i\in I}$, where $I = X$, and for all $i \in I$, $V_i = \{y \mid (i, y) \in E\}$.

Therefore, if E' is a perfect matching of G, the function $R : I \to \bigcup_{i\in I} V_i$, defined as $R(i) = y$, where y is the unique element in V_i such that $(i, y) \in E'$, is an SDR of $\{V_i\}_{i\in I}$.

Theorem 2 (Perfect matching associated to a collection of sets)**.** *Let $\{S_i\}_{i\in I}$ be a collection of non necessarily distinct subsets of an arbitrary set S.*

The directed bipartite digraph associated to $\{S_i\}_{i\in I}$ is built as the graph $G = \langle X, Y, E\rangle$ where $X = I$, $Y = \bigcup_{i\in I} S_i$ and $E = \{(i, x) \mid i \in I \text{ and } x \in S_i\}$.

Therefore, if R is an SDR of $\{S_i\}_{i\in I}$, then the subset of arcs $E' = \{(i, x) \mid i \in I \text{ and } x = R(i)\}$ is a perfect matching of G.

3.1.1 Preliminaries and definitions

The Isabelle Archive of Formal Proofs contains a collection of theories regarding Graph Theory [19]. In particular, Noschinski and Neumann specified, in the theory *Digraph.thy*, the basic data structure *pre_digraph* as the basis to develop complex formalisations such as Kuratowski theorem and the existence of a Eulerian path on directed finite graphs. We also apply such a *record* to establish our formalisation.

```
record ('a,'b) pre_digraph =
  verts :: "'a set"
  arcs :: "'b set"
  tail :: "'b ⇒ 'a"
  head :: "'b ⇒ 'a"
```

Such a record from the theory mentioned above is used since the formalisation established in [19] contains specialised concepts intrinsic to the specific results formalised in it. For example, in the Isabelle AFP theory *Kuratowski.thy* and *complete bipartite digraphs* are defined. However, there is no general specification of bipartite digraphs that are not complete. Consequently, a small variety of basic concepts for graphs were specified. For instance, specifications of the *neighbourhood* of a vertex and the notion of *bipartite_digraph*, among others, are necessary to our development. In the following, some preliminary definitions are presented that were specified to establish the equivalence between the infinite versions of Hall's Theorem.

Arcs of a graph G have tails and heads in the set of vertices of the graph. The binary predicate `neighbour` on pairs of vertices u, v, holds if there exist and arc (u, v) or (v, u) in the graph. A `bipartite_digraph` is a *pre_digraph* G with two disjoint sets of vertices X

and Y, whose union is the set of vertices of the graph, and such that all arcs in the graph have tails in X and heads in Y or vice versa.

```
definition tails:: "('a,'b) pre_digraph ⇒ 'a set" where
  "tails G ≡  { tail G e |e. e ∈ arcs G }"

definition tails_set :: "('a,'b) pre_digraph ⇒ 'b set ⇒ 'a set" where
  "tails_set G E ≡  { tail G e |e. e ∈ E ∧ E ⊆ arcs G }"

definition heads:: "('a,'b) pre_digraph ⇒ 'a set" where
  "heads G ≡  { head G e |e.  e ∈ arcs G }"

definition heads_set:: "('a,'b) pre_digraph ⇒ 'b set ⇒ 'a set" where
  "heads_set G E ≡  { head G e |e.  e ∈ E ∧ E ⊆ arcs G }"

definition neighbour::  "('a,'b) pre_digraph ⇒ 'a ⇒ 'a ⇒ bool" where
  "neighbour G v u  ≡
  ∃e. e∈ (arcs G) ∧ ((head G e = v ∧ tail G e = u) ∨
  (head G e  = u ∧ tail G e = v))"

definition neighbourhood:: "('a,'b) pre_digraph ⇒ 'a  ⇒ 'a set" where
  "neighbourhood G v ≡ {u |u. neighbour G u v}"

definition bipartite_digraph:: "('a,'b) pre_digraph ⇒ 'a set ⇒ 'a set ⇒ bool" where
  "bipartite_digraph G X Y  ≡
     (X ∪ Y = (verts G)) ∧  X ∩ Y = {} ∧
     (∀e ∈ (arcs G).(tail G e) ∈ X ⟷ (head G e) ∈ Y)"
```

The specialised notion of directed bipartite digraphs used is specified in definition dir_bipartite_digraph. Such a graph is a bipartite digraph, consisting of a bi-partition of vertices X and Y in which all *arcs* have *tails* in the set X and *heads* in the set Y. Arcs with the same tail and head are equal.

```
definition dir_bipartite_digraph:: "('a,'b) pre_digraph ⇒ 'a set ⇒ 'a set ⇒ bool"
 where
 "dir_bipartite_digraph G X Y  ≡
        (bipartite_digraph G X Y) ∧  ((tails G = X) ∧
        (∀e1 ∈ arcs G. ∀e2 ∈ arcs G. e1 = e2 ⟷
         head G e1 = head G e2 ∧ tail G e1 = tail G e2))"
```

A matching in a directed bipartite digraph G is specified, in definition dirBD_matching, as a subset E of the arcs of the graph, such that any pair of distinct arcs in E have neither the same head nor the same tail. A perfect matching, specified in definition dirBD_perfect_matching, is a matching in the digraph G that covers the set of vertices X.

```
definition dirBD_matching:: "('a,'b) pre_digraph ⇒ 'a set ⇒ 'a set ⇒ 'b set ⇒ bool"
 where
 "dirBD_matching G X Y E  ≡
        dir_bipartite_digraph G X Y ∧ (E  ⊆  (arcs G)) ∧
        (∀e1∈E. (∀e2∈ E.  e1 ≠ e2 ⟶
        ((head G e1) ≠ (head G e2)) ∧
```

```
        ((tail G e1) ≠ (tail G e2))))"
```

```
definition dirBD_perfect_matching::
  "('a,'b) pre_digraph ⇒ 'a set ⇒ 'a set ⇒ 'b set ⇒ bool"
  where
  "dirBD_perfect_matching G X Y E ≡
   dirBD_matching G X Y E ∧ (tails_set G E = X)"
```

All definitions presented in this subsection belong to the theory `min_graphs` that specialised the concepts of graphs according to the requirements of the target formalisation.

3.1.2 Building SDRs from perfect matchings

Theorem 1 is specified as theorem `dir_BD_to_Hal` below. It uses the definition `E_head` that for any set of arcs E in a digraph and any vertex x, tail of some arc in E, selects the head, y, of an arc in E with tail x. The theorem states that for any directed bipartite digraph, $G = \langle X, Y, E\rangle$ with a perfect matching $E' \subseteq E$, the arcs of G, the family of sets given by the neighbourhoods of vertices $x \in X$ in G, $\{N(x)\}_{x\in X}$, the set of indices given by the set of vertices in X, and the representatives given by *E_head* using the perfect matching E', is an SDR. Since E' is assumed to be a perfect matching, it contains a unique arc for each $x \in X$. Therefore, a unique arc exists with tail x in E'.

The vital required property on *E_head* over matchings and perfect matchings on directed bipartite digraphs is that the operator *E_head* over matchings gives an injective function, and, over perfect matchings, an injective function on X, which is stated as a crucial lemma `dirBD_matching_inj_on`. The proof requires proving a chain of auxiliary lemmas. Among them, a lemma stating the unicity of the operator *E_head* over matchings; and then the construction of an injective function that univocally maps tails into heads on the set of arcs E'.

Using these properties, after unfolding definitions, it is possible to conclude that *(E_head G E)*, as an injective function on X, gives an SDR for the family of neighbourhoods of vertices in X, $\{N(X)\}_{x\in X}$, built from the graph G and the perfect matching E'.

```
definition E_head :: "('a,'b) pre_digraph ⇒ 'b set  ⇒ ('a  ⇒ 'a)"
  where
  "E_head G E = (λx. (THE y. ∃ e. e ∈ E ∧ tail G e = x ∧  head G e = y))"
```

```
theorem dir_BD_to_Hall:
   "dirBD_perfect_matching G X Y E ⟶
   system_representatives (neighbourhood G) X (E_head G E)"
```

3.1.3 Building perfect matchings from SDRs

The structures below are vital to constructing a bipartite digraph from an indexed family of sets.

First, the definition *type1* is used to build the set of vertices in X from the indexation set I; afterward, the *type2* is used to build the set of elements in the set indexed by each $i \in I$, and so obtaining the set of vertices in Y as their union; and, finally, the *type3* is used to build the arcs from all vertices in the set build with *type1* to the union of vertices of sets of *type2*. Moreover, the definition `SDR_bipartite_digraph` uses these three types to reduce a family of sets, indexed by I, on S into a directed bipartite digraph.

```
definition type1:: "'a set ⇒ ('a + 'b) set" where
  "type1 A ≡ {(Inl x)|x.  x ∈ A}"

definition type2:: "'b set ⇒ ('a + 'b) set" where
  "type2 A ≡ {(Inr x)|x.  x ∈ A}"

definition type3:: "'a ⇒ 'b ⇒ ('a + 'b)×('a + 'b)" where
  "type3 x y ≡ ((Inl x), (Inr y))"

definition SDR_bipartite_digraph::
  "('a ⇒ 'b set) ⇒ 'a set ⇒ (('a + 'b), ('a + 'b)×('a + 'b)) pre_digraph"
  where
  "SDR_bipartite_digraph S I ≡
   (| verts = (type1 I) ∪ (⋃i∈ I. type2 ( S i)),
    arcs = {type3 i x |i x. i ∈ I ∧ x ∈ ( S i)},
    tail = (λ(x,y). x),
    head = (λ(x,y). y)
   |)"
```

Theorem 2 is specified as theorem `Hall_to_dir_BD` below. It states that if one has an SDR R for a family of sets, indexed by I, on S, then the associated directed bipartite digraph built using definition *(SDR_bipartite_digraph S I)* has as a perfect matching.

```
theorem Hall_to_dir_BD:
 "system_representatives S I R ⟶
    (dirBD_perfect_matching (SDR_bipartite_digraph S I)
    (type1 I) (⋃i∈ I. type2 ( S i)) {type3 i (R i) |i. i ∈ I})"
```

After unfolding definitions, to prove that this construction indeed satisfies the definition *dirBD_perfect_matching*, it is necessary to prove that one has a matching that covers the set of vertices *type1 I*; i.e., that

$$\begin{array}{ll} \textit{dirBD\_matching (SDR\_bipartite\_digraph S I)} & \textit{(type1 I)} \\ & (\bigcup_{i\in I}.\ \textit{type2 (S i)}) \\ & \{\textit{type3 i (R i)}|i.\ i\in I\} \end{array}$$

and

$$\textit{tails\_set (SDR\_bipartite\_digraph S I)}\{\textit{type3 i (R i)}|i.\ i\in I = \textit{type1 I}\}$$

Both results are achieved by technical lemmas included in the accompanying Isabelle/HOL development.

The latter result, formalised as lemma `SDR_coverage`, is obtained unfolding definitions to see that, in fact, the arcs in $\{\textit{type3 i (R i)} \mid i.\ i \in I\}$ have tails in the set of vertices *(type I)*, and since R is an SDR, it covers all vertices in *(type1 I)*.

The former lemma, formalised in the Isabelle/HOL theory as the elaborated lemma `SDR_dirBD_matching`, requires proving that the construction *(SDR_bipartite_digraph S I)* indeed holds definition *dir_bipartite_digraph* of being a bipartite digraph with arcs directed from the set of vertices *(type1 I)* to the set of vertices $(\bigcup_{i\in I}.\ \textit{type2 (S i)})$, which is also guaranteed by another technical lemma called `SDR_dir_bipartite_digraph`. In addition, by the assumption that R is an SDR, proving that the set of arcs given by $\{\textit{type3 i (R i)} \mid i.\ i \in I\}$ is a matching, requires proving that this set is a subset of the arcs of that one of the construction *(SDR_bipartite_digraph S I)* such that different arcs have neither the same tail nor the same head.

3.2 Formalising the graph-theoretical version of Hall's theorem

Here we explain how the graph-theoretical version of Hall's theorem is obtained from its set-theoretical version formalised in [25]. The graph-theoretical version is stated as Theorem 3.

Theorem 3 (Graph-theoretical version of Hall's Theorem)**.** *Let* $G = \langle X, Y, E \rangle$ *be a directed bipartite digraph.* G *contains a perfect matching covering the set of vertices* X *if and only if*

$$|J| \leq |N(J)| \quad \textit{for all} \quad J \subseteq X$$

This theorem is usually stated for finite graphs only. Also, in contrast to proofs presented in classical textbooks on (finite) graph theory (e.g., [29], [5]), its formalisation, given as the theorem *Hall_digraph* at the end of this section, applies the combinatorial set-theoretical version of this theorem, obtained through application of the compactness theorem for propositional logic, extended for countable sets and published in [25].

The formalisation of this result uses the Theorem 1 proved in Isabelle/HOL as described in Subsection 3.1.2 as theorem *dir_BD_to_Hall* and that states the correctness of the reduction of a directed bipartite digraph $G = \langle X, Y, E \rangle$ with a perfect matching E, to the family of sets of neighbourhoods of vertices X, concluding that the operator *E_head* indeed builds an SDR from the perfect matching E.

The formalisation is based on applying two auxiliary lemmas relating the marriage condition for directed bipartite digraphs to perfect matchings.

The first auxiliary, lemma `marriage_necessary_graph`, states that if a directed bipartite graph has a perfect matching, *(dir_bipartite_digraph G X Y E)*, then the marriage condition holds. Notice that this lemma holds for graphs of arbitrary possible infinite cardinality. Furthermore, relaxing the restriction on countable families to infinite families is possible since the lemma is proved as a consequence of the mechanisation of the fact that the existence of an SDR for arbitrarily infinite indexed families of finite sets implies the marriage condition. The last result was formalised through the theorem `marriage_necessity`, part of the mechanisation reported in [25].

```
lemma marriage_necessary_graph:
assumes "(dirBD_perfect_matching G X Y E)" and
        "∀ i∈X. finite (neighbourhood G i)"
shows "∀ J⊆X. finite J ⟶ (card J) ≤ card (⋃ (neighbourhood G ‘ J))"
```

The tricky part of this lemma is applying the transformation *(system_representatives (neighbourhood G) X (E_head G E))* through theorem *dir_BD_to_Hall* in order to obtain from the SDR, an injective function R from any subset J to their representatives in the union of neighbourhoods of elements $j \in J$ such that: *card* $J \leq$ *card* $(\bigcup_{j \in J} N(j))$. The injectivity of R, guaranteed by theorem *dir_BD_to_Hall*, implies the desired inequation.

The second auxiliary lemma, `marriage_sufficiency_graph` below, states that if the marriage condition holds for a countable directed bipartite graph, then there exists a perfect matching.

```
lemma marriage_sufficiency_graph:
```

```
fixes  G :: "('a, 'b) pre_digraph" and X:: "'a set"
assumes "dir_bipartite_digraph G X Y"  and "∀ i∈X. finite (neighbourhood G i)"
and "∃g. enumeration (g:: nat ⇒ 'a)" and  "∃h. enumeration (h:: nat ⇒ 'b)"
shows
  "(∀ J⊆X. finite J ⟶ (card J) ≤ card (⋃ (neighbourhood G ` J))) ⟶
  (∃E. dirBD_perfect_matching G X Y E)"
```

This lemma applies the formalisation of the countable set-theoretical version of Hall's theorem ([25]) to infer the existence of an SDR R for the countable indexed family of sets $\{N(i)\}_{i\in X}$. Applying the lemma is possible since the marriage condition for this family of sets is the premise of the target implication. From the system of representatives, it is possible to build the perfect matching as the set of arcs $\{(i, R(i))\}_{i\in X}$. Through two additional auxiliary lemmas, it is proved that this set covers the set of vertices X (lemma `perfect` ↗) and is indeed a matching (lemma `dirBD_matching` ↗). Therefore, one concludes that *(dirBD_perfect_matching* G X Y $\{(i, R(i))\}_{i\in X}$).

Finally, the countable graph-theoretical version of Hall's theorem, specified as theorem `Hall_digraph` ↗, is formalised as below. The use of necessity and sufficiency auxiliary lemmas is highlighted in the mechanisation.

```
theorem Hall_digraph:
fixes  G :: "('a, 'b) pre_digraph" and X:: "'a set"
assumes "dir_bipartite_digraph G X Y"  and "∀ i∈X. finite (neighbourhood G i)"
and  "∃g. enumeration (g:: nat ⇒ 'a)" and  "∃h. enumeration (h:: nat ⇒ 'b)"
shows "(∃E. dirBD_perfect_matching G X Y E) ⟷
        (∀ J⊆X. finite J ⟶ (card J) ≤ card (⋃ (neighbourhood G ` J)))"
proof
  assume hip1:  " ∃E. dirBD_perfect_matching G X Y E"
  show  "(∀ J⊆ X. finite J ⟶ (card J) ≤ card (⋃  (neighbourhood G ` J)))"
    using hip1 assms(1-2) marriage_necessary_graph[of G X Y] by auto
next
  assume hip2: "∀ J⊆ X. finite J ⟶ card J ≤ card (⋃  (neighbourhood G ` J))"
  show "∃E. dirBD_perfect_matching G X Y E"
    using assms  marriage_sufficiency_graph[of G X Y] hip2
  proof-
    have "(∀ J⊆ X. finite J ⟶ (card J) ≤ card (⋃  (neighbourhood G ` J))) ⟶
          (∃E. dirBD_perfect_matching G X Y E)"
      using assms marriage_sufficiency_graph[of G X Y] by auto
    thus ?thesis using hip2 by auto
  qed
qed
```

4 Related Work

Extensions to the infinite case from theorems equivalent to Hall's marriage theorem in the finite case are generally not straightforward. In addition to the infinite version of Hall's marriage theorem, our development includes formalisations of infinite versions of De Bruijn-Erdös graph colouring theorem ([6]), and König lemma ([17]), obtained from the compactness theorem for predicate logic (theorems available through the links ↗ and ↗, respectively). Moreover, even such extensible theorems would not necessarily be provable from the compactness theorem and elementary techniques. An example is König's duality theorem, proved by Aharoni [1], and subsequently studied in detail by Aharoni et al. [2]. This theorem states that in every bipartite graph $G = \langle X, Y, E \rangle$, *there exists* a matching $M \subseteq E$ such that selecting one vertex from each arc in M one has a cover of the graph. König duality theorem is a strong form of the finite well-known König-Egerváry theorem that states that in a finite bipartite graph, the size of a maximal matching is equal to the size of a minimal cover [18]. The vital difference of the duality theorem is that such a cover of the graph cannot be extracted from an arbitrary matching. Indeed, from a matching, it is possible to build a cover of the same cardinality as the cardinality of the matching, but not that it covers the graph. So, the notion of *König cover* came to arise, which is defined as a cover of the graph that consists of a selection of one vertex from each arc of a matching.

Lifting results from the finite to the infinite through the application of compactness (of König's lemma) corresponds to a recursive construction of a procedure that produces the target solution in the degree of unsolvability of the halting problem [2]. Such a recursive construction is possible for Dilworth's theorem (restricting the maximal anti-chains in infinite partial ordered sets to be finite - [7], see also Sec. 2.5 in [14]) but not for König's duality theorem. Indeed, Aharoni et al. [2] proved that the complexity of constructing covers exceeds the complexity of the halting problem, it is even a problem of higher complexity than answering all first-order questions about arithmetic. Also, they proved that the compactness theorem and König's lemma do not suffice to prove the duality theorem and other related results in matching theory.

The first formalisation of the finite version of Hall's Theorem was developed in Mizar by Romanowicz and Grabowski [22]. Also, there are formalisations in Isabelle/HOL by Jiang and Nipkow [16]. Both these formalisations follow Rado's proof [20], but the last one also includes a mechanisation based on Halmos and Vaughan's proof [13]. In addition, there is a formalisation in Coq that uses formalisations of Dilworth's decomposition theorem and bi-partitions in graphs [26]. An earlier formalisation of Dilworth's theorem in Mizar is presented in [23]. Recently, Gusakov, Mehta and Miller [11] presented three different proofs of the finite version of Hall's theorem in Lean in terms of indexed families of finite subsets, of the existence of injections that saturate binary relations over finite sets, and of matchings in bipartite graphs. Related combinatorial results are reported in recent works by Doczkal et al. in their graph theory Coq library (e.g., [8], [10], and [9]). Additionally, Singh and Natarajan formalised in Coq other combinatorial results as the perfect graph theorem and a weak version of this theorem (e.g., [27], [28]).

Known mechanisations of the enumerable version of the set-theoretical version of Hall's theorem appear in the formalisation used in the authors' work, previously discussed, [25], and in Gusakov, Mehta, and Miller's work [11]. The former work uses the compactness theorem for predicate logic. In the latter work, the authors apply an *inverse limit* version of the König's lemma. This lemma states that if $\{X_i\}_{i\in\mathbb{N}}$ is an indexed family of nonempty finite sets with functions $f_i : X_{i+1} \to X_i$, for each $i \in \mathbb{N}$, then there exists a family of elements $x \in \prod_i X_i$ such that $x_i = f_i(x_{i+1})$, for all $i \in \mathbb{N}$. König's lemma follows from this infinite limit version by choosing as set X_i the paths of length i from the root vertex v_0 in a tree. So, the function f_i maps paths in X_{i+1} into the paths without their last arc that are paths that belong to X_i. The inverse limit consists of the infinite chain of functions $f_1, f_2, \ldots$. König's lemma is applied to prove the enumerable version of Hall's theorem by taking M_n as the set of all matchings on the first n indices of I (i.e., the set of all possible SDRs for the sets $S_1, \ldots, S_n$), and $f_n : M_{n+1} \to M_n$ as the restriction of a matching to a smaller set of indices. Since the marriage condition holds for the finite indexed families, each M_n is nonempty, and by König's lemma, an element of the inverse limit gives a matching on I.

5 Conclusions and Future Work

This paper presented the formalisation in Isabelle/HOL of the graph-theoretical version of Hall's theorem for countable (infinite) graphs. The prominent feature of the formalisation is following a presentation close to pen-and-paper proofs. This development will enable other mechanisations of infinite combinatorial, set-theoretical, and graph-theoretical results related to the compactness theorem for predicate logic and its derivations, König lemma, Hall's marriage theorem, and de Bruijn-Erdös k-colouring theorem, such as generalisations of Dilwort's theorem.

An exciting challenge for future research consists in developing the required formal background in proof assistants to enable the formalisation of other theorems which do not extend straightforwardly from the results mentioned above, such as the König duality theorem, among others.

References

[1] Ron Aharoni. König's Duality Theorem for Infinite Bipartite Graphs. *Journal of the London Mathematical Society*, s2-29(1), 1984. https://doi.org/10.1112/jlms/s2-29.1.1.

[2] Ron Aharoni, Menachem Magidor, and Richard A Shore. On the Strength of König's Duality Theorem for Infinite Bipartite Graphs. *Journal of Combinatorial Theory Series B*, (54):257–290, 1992. https://doi.org/10.1016/0095-8956(92)90057-5.

[3] Robert D. Borgersen. Equivalence of seven major theorems in combinatorics, 2004. Talk available at Department of Mathematics, University of Manitoba, Canada. https://home.cc.umanitoba.ca/~borgerse/Presentations/GS-05R-1.pdf.

[4] Peter J. Cameron. *Combinatorics: Topics, Techniques, Algorithms*. Cambridge University Press, 1994.

[5] Gary Chartrand, Linda Lesniak, and Ping Zhang. *Graphs and Digraphs*. Chapman & Hall/CRC, 5th edition, 2010.

[6] Nicolaas Govert De Bruijn and Pál Erdös. A Colour Problem for Infinite Graphs and a Problem in the Theory of Relations. *Indagationes Mathematicae (Proceedings)*, 54:371–373, 1951.

[7] Robert P. Dilworth. A Decomposition Theorem for Partially Ordered Sets. *Annals of Mathematics*, 51(1):161–166, 1950. https://doi.org/10.2307/1969503.

[8] Christian Doczkal, Guillaume Combette, and Damien Pous. A Formal Proof of the Minor-Exclusion Property for Treewidth-Two Graphs. In *Proceedings 9th International Conference on Interactive Theorem Proving - ITP*, volume 10895 of *Lecture Notes in Computer Science*, pages 178–195. Springer, 2018. https://doi.org/10.1007/978-3-319-94821-8_11.

[9] Christian Doczkal and Damien Pous. Completeness of an axiomatization of graph isomorphism via graph rewriting in Coq. In *Proceedings of the 9th ACM SIGPLAN International Conference on Certified Programs and Proofs - CPP*, pages 325–337. ACM, 2020.

[10] Christian Doczkal and Damien Pous. Graph Theory in Coq: Minors, Treewidth, and Isomorphisms. *Journal of Automated Reasoning*, 64(5):795–825, 2020. https://doi.org/10.1007/s10817-020-09543-2.

[11] Alena Gusakov, Bhavik Mehta, and Kyle A. Miller. Formalizing Hall's Marriage Theorem in Lean. *arXiv abs/2101.00127[math.CO]*, 2021. https://doi.org/10.48550/arxiv.2101.00127.

[12] Philip Hall. On representatives of subsets. *London Mathematical Society*, 10:26–30, 1935. https://doi.org/10.1112/jlms/s1-10.37.26.

[13] Paul R. Halmos and Herbert E. Vaughan. The Marriage Problem. *American Journal of Mathematics*, 72(1):214–215, 1950.

[14] Egbert Harzheim. *Ordered Sets*, volume 7 of *Advances in Mathematics*. Springer, 2005.

[15] Dongchen Jiang and Tobias Nipkow. Hall's Marriage Theorem. *Archive of Formal Proofs*, 2010, 2010.

[16] Dongchen Jiang and Tobias Nipkow. Proof Pearl: The Marriage Theorem. In *Proceedings First International Conference on Certified Programs and Proofs - CPP*, volume 7086 of *Lecture Notes in Computer Science*, pages 394–399, 2011. https://doi.org/10.1007/978-3-642-25379-9_28.

[17] Dénes König. Über eine Schlussweise aus dem Endlichen ins Unendliche. *Acta Sci. Math. (Szeged)*, 3(2-3):121–130, 1927.

[18] Dénes König. *Theorie Der Endlichen und Unendlichen Graphen: Kombinatorische Topologie Der Streckenkomplexe*, volume 16 of *Mathematik und ihre Anwendungen in Monographien und Lehrbüchern*. Chelsea, 1936.

[19] Lars Noschinski. Graph theory. *Archive of Formal Proofs*, 2013. http://isa-afp.org/entries/Graph_Theory.html, Formal proof development.

[20] Richard Rado. Note on the transfinite case of Hall's theorem on representatives. *London Mathematical Society*, S1-42(1):321–324, 1967. https://doi.org/10.1112/jlms/s1-42.1.321.

[21] Philip F. Reichmeider. *The equivalence of some combinatorial matching theorems*. Polygonal Publishing House, 1985.

[22] Ewa Romanowicz and Adam Grabowski. The Hall Marriage Theorem. *Formalized Mathematics (University of Białystok)*, 12(3):315–320, 2004. https://fm.mizar.org/2004-12/pdf12-3/hallmar1.pdf.

[23] Piotr Rudnicki. Dilworth's Decomposition Theorem for Posets. *Formalized Mathematics*, 17(4):223–232, 2009. https://doi.org/10.2478/v10037-009-0028-4.

[24] Fabián Fernando Serrano Suárez. *Formalización en Isar de la Meta-Lógica de Primer Orden*. PhD thesis, Departamento de Ciencias de la Computación e Inteligencia Artificial, Universidad de Sevilla, Spain, 2012. https://idus.us.es/handle/11441/57780. In Spanish.

[25] Fabián Fernando Serrano Suárez, Mauricio Ayala-Rincón, and Thaynara Arielly de Lima. Hall's Theorem for Enumerable Families of Finite Sets. In *Proc. Int. Conf. on Intelligent Computer Mathematics CICM*, pages 107–121. Springer, 2022. http://doi.org/10.1007/978-3-031-16681-5_7.

[26] Abhishek Kr Singh. Formalization of some central theorems in combinatorics of finite sets. *arXiv abs/1703.10977[cs.Lo]*, 2017. Short presentation at the 21st International Conference on Logic for Programming, Artificial Intelligence and Reasoning - LPAR. https://doi.org/10.48550/arxiv.1703.10977.

[27] Abhishek Kr Singh and Raja Natarajan. Towards a Constructive Formalization of Perfect Graph Theorems. In *Proceedings 8th Indian Conference on Logic and Its Applications - ICLA*, volume 11600 of *Lecture Notes in Computer Science*, pages 183–194. Springer, 2019. https://doi.org/10.1007/978-3-662-58771-3_17.

[28] Abhishek Kr Singh and Raja Natarajan. A constructive formalization of the weak perfect graph theorem. In *Proceedings of the 9th ACM SIGPLAN International Conference on Certified Programs and Proofs - CPP*, pages 313–324. ACM, 2020. https://doi.org/10.1145/3372885.3373819.

[29] Douglas Brent West. *Introduction to Graph Theory*. Pearson Modern Classics for Advanced Mathematics. Pearson Education, Inc, second edition, 2001.